
Alternative Time Metrics: Age, Death, or Time in Study? 

COMPLETED VERSION 
 
SAS Code: 
 
* Output options: all can be turned on/off by adding or removing the 'NO': 
  page number, date, centering, or page breaks, page length and width, 2-digit year code; 
  OPTIONS NOnumber NOdate NOcenter FormDlim=' ' PageSize=MAX LineSize=MAX YEARCUTOFF=2000; 
* Eliminate SAS default titles and names of tables in output (TRACE ON to show); 
  TITLE; ODS TRACE OFF; 
 
* Defining global variable for datafile name to be replaced in code below; 
%GLOBAL udatafile filesave;  
 %LET udatafile= OctoAll;  
 %LET filesave= F:\Example Data\OCTO; 
 
* Location for SAS files for these models (pulls from above); 
LIBNAME octo "&filesave."; 
 
* Importing data into work library and centering predictors; 
DATA &udatafile.; SET octo.&udatafile.; 
 
 * Creating time-in-study variable (compage = current age); 
  time = compage - compage1;   
   LABEL time= "time: Time in Study (0=Baseline)"; 
  roundtime = ROUND(time, 2);    
   LABEL roundtime= "roundtime: Time Rounded to 2 Years"; 
 
 * Creating age variables for analysis; 
  agecohort85  = compage1 - 85;  
   LABEL agecohort85= "agecohort85: Age at Entry (0=85)"; 
  agetime85 = compage - 85;      
   LABEL agetime85= "agetime85: Current Age (0=85)"; 
 * Creating rounded age variable for PLOTTING ONLY; 
  roundage = ROUND(compage, 2);  
   LABEL roundage= "roundage: Age Rounded to 2 Years"; 
 
 * Creating death variables for analysis; 
 IF (dead=1) THEN DO; * Calculate only for dead people; 
  deadcohort6 = compage1 - DeadAge + 6;  
   LABEL deadcohort6= "deadcohort6: Years to Death from Entry (0=6)"; 
  deadtime  = compage - DeadAge;      
   LABEL deadtime= "deadtime: Current Years to Death"; 
  deadtime6 = compage - DeadAge + 6;     
   LABEL deadtime6= "deadtime6: Current Years to Death (0=6)"; 
 * Creating rounded years to death variable for PLOTTING ONLY; 
  rounddead = ROUND(deadtime, 2);  
   LABEL rounddead= "rounddead: YTD Rounded to 2 Years"; 
 END;  
RUN; 
 
* Including only cases that have age and years to death at baseline; 
DATA &udatafile.; SET &udatafile.; 
 WHERE NMISS(agecohort85, deadcohort6)=0; 
RUN; 

 
 
Empty Model Results: 
 
ICC for time-varying prose recall (our outcome) = .40  
 
ICC for time-varying age (a predictor) = .52 
ICC for time-varying years to death (a predictor) = .25 

Age as Time: Examining Means across Ages  
 
* Setting system options for graphics below; 
GOPTIONS RESET=ALL NOBORDER FTEXT=TRIPLEX FTITLE=TRIPLEX VSIZE=4.0in HSIZE=5.0in INTERPOL=JOIN  
GSFNAME=OUTGRAPH DEV=BMP GSFMODE=REPLACE; 
 
TITLE1 "SAS Means Model for Prose Recall by Age"; 
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=ML; 
 CLASS case roundage; 
 MODEL prose = roundage / SOLUTION DDFM=Satterthwaite; 
 RANDOM INTERCEPT / G TYPE=UN SUBJECT=case;  
 LSMEANS roundage;  
 ODS OUTPUT LSMeans=AgeMeans; RUN; 
 
* Opening up saved data set of means;  
DATA AgeMeans2; SET AgeMeans; 
 case=-99; compage=roundage; prose=estimate; 
 KEEP case compage prose;  
RUN; 
 
* Good practice -- sort datasets prior to merging; 
PROC SORT DATA=&udatafile.; BY case compage; RUN; 
PROC SORT DATA=AgeMeans2;   BY case compage; RUN; 
DATA AgeMerged; SET &udatafile. AgeMeans2; BY case compage; RUN; 
 
* Generating plot for INDIVIDUALS and MEAN, saving it as bitmap to named location; 
FILENAME outgraph "&filesave.\Prose by Age.bmp";  
TITLE JUSTIFY=CENTER HEIGHT=1.5 "Prose Recall by Age";  
PROC GPLOT DATA=AgeMerged;  
 * Settings for Y-axis; 
    AXIS1 LENGTH=3.0in LABEL=(ANGLE=90 HEIGHT=1.2 "Prose Recall") ORDER=(-2 TO 16 BY 2); 
 * Settings for X-axis; 
    AXIS2 LENGTH=4.0in LABEL=(HEIGHT=1.2 "Age") ORDER=(80 TO 98 BY 2); 
  * Specification for observed means; 
    SYMBOL1 VALUE=DOT COLOR=BLACK HEIGHT=1.5 LINE=1 WIDTH=5; 
  * Specification for all other lines (from REPEAT option); 
    SYMBOL2 VALUE=NONE LINE=1 WIDTH=1 REPEAT=100; 
  * Plot Y*X=separate person, start axes at 0 unless otherwise specified; 
    PLOT prose*compage=case / VZERO HZERO NOLEGEND VAXIS=AXIS1 HAXIS=AXIS2;   
RUN; QUIT; 
 
 

 
 
 

Age 
LS Means 
Estimate 

LS Means
SE 

80 7.540  0.626

82 6.201  0.421

84 5.180  0.408

86 4.342  0.405

88 3.154  0.456

90 1.855  0.532

92 1.247  0.730

94 ‐0.334  1.061

96 ‐1.253  1.395

98 ‐0.769  1.941

1a) Age as Time: Random Linear Model 
 
 TITLE1 "SAS Random Linear Model for Age as Time"; 
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=ML; 
 CLASS case; 
 MODEL prose = agetime85 / SOLUTION DDFM=Satterthwaite; 
 RANDOM INTERCEPT agetime85 / G TYPE=UN SUBJECT=case; RUN; 
 
                   Estimated G Matrix                      
 Row    Effect       Case              Col1        Col2 
   1    Intercept      3            10.3759     -0.4298 
   2    agetime85      3            -0.4298      0.1214 
 
                  Covariance Parameter Estimates 
                                    Standard         Z 
Cov Parm     Subject    Estimate       Error     Value        Pr Z 
UN(1,1)      Case        10.3759      1.5551      6.67      <.0001 
UN(2,1)      Case        -0.4298      0.2088     -2.06      0.0395 
UN(2,2)      Case         0.1214     0.07364      1.65      0.0496 
Residual                  7.3771      0.8104      9.10      <.0001 
 
           Fit Statistics 
-2 Log Likelihood              2540.3 
AIC (smaller is better)        2552.3 
AICC (smaller is better)       2552.5 
BIC (smaller is better)        2572.8 
 
                   Solution for Fixed Effects 
                         Standard 
Effect       Estimate       Error      DF    t Value    Pr > |t| 
Intercept      4.7296      0.2847     178      16.61      <.0001 
agetime85     -0.5134     0.05938    71.2      -8.65      <.0001  
 

How do we interpret the fixed intercept? 
The fixed intercept of 4.7 is the expected prose recall for an 85-year-old. 
 
How do we interpret the agetime85 linear slope? 
For every year above 85, prose recall decreases by -.51. 
 
 
1b) Examining Convergence of Between-Person and Within-Person Effects of Age  
 
TITLE1 "SAS Testing Age Convergence"; 
TITLE2 "Effect of Age Cohort on Intercept and Linear Change"; 
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=ML; 
 CLASS case; 
 MODEL prose = agetime85 agecohort85 agetime85*agecohort85 / SOLUTION DDFM=Satterthwaite; 
 RANDOM INTERCEPT agetime85 / G TYPE=UN SUBJECT=case; RUN; 
 
                  Covariance Parameter Estimates 
                                    Standard         Z 
Cov Parm     Subject    Estimate       Error     Value        Pr Z 
UN(1,1)      Case        10.1450      1.4945      6.79      <.0001 
UN(2,1)      Case        -0.3190      0.2096     -1.52      0.1280 
UN(2,2)      Case         0.1374     0.07666      1.79      0.0366 
Residual                  7.0297      0.7639      9.20      <.0001 
 
           Fit Statistics 
-2 Log Likelihood              2530.2 
AIC (smaller is better)        2546.2 
AICC (smaller is better)       2546.5 
BIC (smaller is better)        2573.5 

                         Solution for Fixed Effects 
                                     Standard 
Effect                   Estimate       Error      DF    t Value    Pr > |t| 
Intercept                  5.5132      0.4155     330      13.27      <.0001 
agetime85                 -0.6752     0.07922    91.5      -8.52      <.0001 
agecohort85                0.4150      0.1321     342       3.14      0.0018 
agetime85*agecohort85    -0.01855     0.02063    75.8      -0.90      0.3715 

 
NOW how do we interpret the fixed intercept? 
The fixed intercept of 4.7 is the expected prose recall for someone who began the study at 85 and is currently 85. 
 
 
NOW how do we interpret the agetime85 linear slope? 
For someone who began the study at 85, for every year above 85 they are, prose recall declines by .67. 
 
 
What does the effect of agecohort85 represent? 
For every year older than 85 when beginning the study, prose recall at age 85 goes up by .42. 
 
 
What does the interaction of agetime85*agecohort85 represent? 
For every year older than 85 when beginning the study, the rate of age decline per year is greater (more 
negative) by .02 (n.s.). 

Years to Death as Time: Examining Means across Years to Death 
 
/*** Plot for Years to Death Means ***/ 
 
* Setting system options for graphics below; 
GOPTIONS RESET=ALL NOBORDER FTEXT=TRIPLEX FTITLE=TRIPLEX VSIZE=4.0in HSIZE=5.0in INTERPOL=JOIN  
GSFNAME=OUTGRAPH DEV=BMP GSFMODE=REPLACE; 
 
TITLE1 "SAS Means Model for Prose Recall by Years to Death"; 
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=ML; 
 CLASS case rounddead; 
 MODEL prose = rounddead / SOLUTION DDFM=Satterthwaite; 
 RANDOM INTERCEPT / G TYPE=UN SUBJECT=case;  
 LSMEANS rounddead;  
 ODS OUTPUT LSMeans=DeadMeans; RUN; 
 
* Opening up saved data set of means;  
DATA DeadMeans2; SET DeadMeans; 
 case=-99; deadtime=rounddead; prose=estimate; 
 KEEP case deadtime prose;  
RUN; 
 
* Good practice -- sort datasets prior to merging; 
PROC SORT DATA=&udatafile.; BY case deadtime; RUN; 
PROC SORT DATA=DeadMeans2; BY case deadtime; RUN; 
DATA DeadMerged; SET &udatafile. DeadMeans2; BY case deadtime; RUN; 
 
* Generating plot for INDIVIDUALS and MEAN, saving it as bitmap to named location; 
FILENAME outgraph "&filesave.\Prose by YTD.bmp";  
TITLE JUSTIFY=CENTER HEIGHT=1.5 "Prose Recall by Years to Death";  
PROC GPLOT DATA=DeadMerged;  
 * Settings for Y-axis; 
    AXIS1 LENGTH=3.0in LABEL=(ANGLE=90 HEIGHT=1.2 "Prose Recall") ORDER=(0 TO 16 BY 2); 
 * Settings for X-axis; 
    AXIS2 LENGTH=4.0in LABEL=(HEIGHT=1.2 "Years to Death") ORDER=(-14 TO 0 BY 2); 
  * Specification for observed means; 
    SYMBOL1 VALUE=DOT COLOR=BLACK HEIGHT=1.5 LINE=1 WIDTH=5; 
  * Specification for all other lines (from REPEAT option); 
    SYMBOL2 VALUE=NONE LINE=1 WIDTH=1 REPEAT=100; 
  * Plot Y*X=separate person, start axes at 0 unless otherwise specified; 
    PLOT prose*deadtime=case / VZERO HZERO NOLEGEND VAXIS=AXIS1 HAXIS=AXIS2;   
RUN; QUIT; 
 
 

 
 

Years  
to Death 

LS Means 
Estimate 

LS Means
 SE 

‐14 7.784  1.482

‐12 9.328  0.837

‐10 7.374  0.591

‐8 6.961  0.447

‐6 6.451  0.383

‐4 4.361  0.341

‐2 2.294  0.340

0 1.179  0.542

2a) Years to Death as Time: Fixed Quadratic, Random Linear Model 
 
TITLE1 "SAS Fixed Quadratic, Random Linear Model for Years to Death as Time"; 
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=ML; 
 CLASS case; 
 MODEL prose = deadtime6 deadtime6*deadtime6 / SOLUTION DDFM=Satterthwaite; 
 RANDOM INTERCEPT deadtime6 / G TYPE=UN SUBJECT=case; RUN; 
                  Covariance Parameter Estimates 
                                    Standard         Z 
Cov Parm     Subject    Estimate       Error     Value        Pr Z 
UN(1,1)      Case         8.6019      1.4991      5.74      <.0001 
UN(2,1)      Case        -0.5511      0.2420     -2.28      0.0228 
UN(2,2)      Case        0.03821     0.05997      0.64      0.2620 
Residual                  7.5763      0.8030      9.43      <.0001 
 
           Fit Statistics 
-2 Log Likelihood              2467.9 
AIC (smaller is better)        2481.9 
AICC (smaller is better)       2482.2 
BIC (smaller is better)        2505.8 
 
                        Solution for Fixed Effects 
                                   Standard 
Effect                 Estimate       Error      DF    t Value    Pr > |t| 
Intercept                6.0177      0.3095     197      19.45      <.0001 
deadtime6               -0.6663     0.06177    61.3     -10.79      <.0001 
deadtime6*deadtime6    -0.04428     0.01364     204      -3.25      0.0014 
 
How do we interpret the fixed intercept? 
The fixed intercept of 6.0 is the expected prose recall at 6 years before death. 
 
How do we interpret the deadtime6 linear slope? 
For every year closer to death than 6, prose recall declines by .67, as evaluated at 6 YTD. 
 
How do we interpret the deadtime6*deadtime6 quadratic slope? 
The rate of decline becomes .044*2 more negative (stronger) for every year closer to death (accelerating decline).  
 
 
2b) Examining Convergence of Between-Person and Within-Person Effects of Years to Death  
 
TITLE1 "SAS Testing Death Convergence"; 
TITLE2 "Effect of Death Cohort on Intercept and Linear Change"; 
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=ML; 
 CLASS case; 
 MODEL prose = deadtime6 deadtime6*deadtime6 deadcohort6  
    deadtime6*deadcohort6 deadtime6*deadtime6*deadcohort6 
   / SOLUTION DDFM=Satterthwaite; 
 RANDOM INTERCEPT deadtime6 / G TYPE=UN SUBJECT=case; RUN; 
 
                  Covariance Parameter Estimates 
                                    Standard         Z 
Cov Parm     Subject    Estimate       Error     Value        Pr Z 
UN(1,1)      Case         8.3628      1.4682      5.70      <.0001 
UN(2,1)      Case        -0.5053      0.2332     -2.17      0.0303 
UN(2,2)      Case        0.03011     0.05808      0.52      0.3021 
Residual                  7.6021      0.8074      9.42      <.0001 
 
           Fit Statistics 
-2 Log Likelihood              2465.8 
AIC (smaller is better)        2485.8 
AICC (smaller is better)       2486.3 
BIC (smaller is better)        2519.9 

                              Solution for Fixed Effects 
                                               Standard 
Effect                             Estimate       Error      DF    t Value    Pr > |t| 
Intercept                            6.3695      0.3901     224      16.33      <.0001 
deadtime6                           -0.8606      0.1785     227      -4.82      <.0001 
deadtime6*deadtime6                -0.01586     0.03389     365      -0.47      0.6400 
deadcohort6                          0.1749      0.1258     252       1.39      0.1657 
deadtime6*deadcohort6              -0.04874     0.03982     215      -1.22      0.2223 
deadtime6*deadtime6*deadcohort6    0.001366    0.004894     212       0.28      0.7805 
 
NOW how do we intercept the fixed intercept, linear deadtime6, and quadratic deadtime6 slopes? 
These are specifically for someone who entered the study 6 years from death. 
 
 
 
What does the effect of deadcohort6 represent? 
For every year closer to death at baseline past 6 years, prose recall at 6 YTD is greater by .17. 
 
 
What do the interactions of deadcohort6 on the linear and quadratic deadtime6 slopes represent? 
For every year closer to death at baseline past 6 years, the linear rate of decline as a function of YTD 
becomes more negative by .05, whereas the quadratic rate of decline as a function of YTD becomes less 
negative by .001. 
 
 
 

 

0

2

4

6

8

10

-9 -7 -5 -3 -1

Years To Death

P
re
d
ic
te
d
 O
u
tc
o
m
e

Initial YTD 9 Initial YTD 7 Initial YTD 5 Death Convergence



Psyc 944 Example 11 page 8 

  

Years in Study as Time: Examining Means across Years in Study 
 
/*** Plot for Years in Study Means ***/ 
 
* Setting system options for graphics below; 
GOPTIONS RESET=ALL NOBORDER FTEXT=TRIPLEX FTITLE=TRIPLEX VSIZE=4.0in HSIZE=5.0in INTERPOL=JOIN  
GSFNAME=OUTGRAPH DEV=BMP GSFMODE=REPLACE; 
 
TITLE1 "SAS Means Model for Prose Recall by Years in Study"; 
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=ML; 
 CLASS case roundtime; 
 MODEL prose = roundtime / SOLUTION DDFM=Satterthwaite; 
 RANDOM INTERCEPT / G TYPE=UN SUBJECT=case;  
 LSMEANS roundtime;  
 ODS OUTPUT LSMeans=TimeMeans; RUN; 
 
* Opening up saved data set of means;  
DATA TimeMeans2; SET TimeMeans; 
 case=-99; time=roundtime; prose=estimate; 
 KEEP case time prose;  
RUN; 
 
* Good practice -- sort datasets prior to merging; 
PROC SORT DATA=&udatafile.; BY case time; RUN; 
PROC SORT DATA=TimeMeans2; BY case time; RUN; 
DATA TimeMerged; SET &udatafile. TimeMeans2; BY case time; RUN; 
 
* Generating plot for INDIVIDUALS and MEAN, saving it as bitmap to named location; 
FILENAME outgraph "&filesave.\Prose by Time.bmp";  
TITLE JUSTIFY=CENTER HEIGHT=1.5 "Prose Recall by Years in Study";  
PROC GPLOT DATA=TimeMerged;  
 * Settings for Y-axis; 
    AXIS1 LENGTH=3.0in LABEL=(ANGLE=90 HEIGHT=1.2 "Prose Recall") ORDER=(0 TO 16 BY 2); 
 * Settings for X-axis; 
    AXIS2 LENGTH=4.0in LABEL=(HEIGHT=1.2 "Years in Study") ORDER=(0 TO 8 BY 2); 
  * Specification for observed means; 
    SYMBOL1 VALUE=DOT COLOR=BLACK HEIGHT=1.5 LINE=1 WIDTH=5; 
  * Specification for all other lines (from REPEAT option); 
    SYMBOL2 VALUE=NONE LINE=1 WIDTH=1 REPEAT=100; 
  * Plot Y*X=separate person, start axes at 0 unless otherwise specified; 
    PLOT prose*time=case / VZERO HZERO NOLEGEND VAXIS=AXIS1 HAXIS=AXIS2;   
RUN; QUIT; 

 
 
 
 
Years in 
Study 

LS Means  
Estimate 

LS Means 
SE 

0 5.760  0.333

2 4.376  0.347

4 2.845  0.417

6 1.531  0.517

8 1.180  0.662

3a) Years in Study as Time: Random Linear Model 
 
TITLE1 "SAS Random Linear Model for Years in Study as Time"; 
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=ML; 
 CLASS case; 
 MODEL prose = time / SOLUTION DDFM=Satterthwaite; 
 RANDOM INTERCEPT time / G TYPE=UN SUBJECT=case; RUN; 

 
                  Covariance Parameter Estimates 
                                    Standard         Z 
Cov Parm     Subject    Estimate       Error     Value        Pr Z 
UN(1,1)      Case        13.7180      2.1259      6.45      <.0001 
UN(2,1)      Case        -0.9713      0.4278     -2.27      0.0232 
UN(2,2)      Case         0.2953      0.1151      2.57      0.0052 
Residual                  6.3442      0.6834      9.28      <.0001 
 
           Fit Statistics 
-2 Log Likelihood              2532.2 
AIC (smaller is better)        2544.2 
AICC (smaller is better)       2544.4 
BIC (smaller is better)        2564.7 
 
                   Solution for Fixed Effects 
                         Standard 
Effect       Estimate       Error      DF    t Value    Pr > |t| 
Intercept      5.6882      0.3261     179      17.44      <.0001 
time          -0.6555     0.07879    83.1      -8.32      <.0001 
 
How do we interpret the fixed intercept? 
The fixed intercept is the expected prose recall on the first occasion of the study (time in study=0) 
 
How do we interpret the time linear slope? 
For every additional year in the study, prose recall declines by .66. 
 
 
 
3b) Examining Prediction by Baseline Age and Years to Death (r = .20 between time metrics) 
 
TITLE1 "SAS Examining Prediction by Age and Years to Death"; 
TITLE2 "Effect of Age Cohort and Death Cohort on Intercept and Linear Change"; 
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=ML; 
 CLASS case; 
 MODEL prose = time agecohort85 deadcohort6 time*agecohort85 time*deadcohort6 
   / SOLUTION DDFM=Satterthwaite; 
 RANDOM INTERCEPT time / G TYPE=UN SUBJECT=case; RUN; 
 
                  Covariance Parameter Estimates 
                                    Standard         Z 
Cov Parm     Subject    Estimate       Error     Value        Pr Z 
 
UN(1,1)      Case        10.1009      1.7098      5.91      <.0001 
UN(2,1)      Case        -0.8857      0.3614     -2.45      0.0143 
UN(2,2)      Case         0.2632      0.1038      2.54      0.0056 
Residual                  6.0535      0.6353      9.53      <.0001 
 
           Fit Statistics 
-2 Log Likelihood              2463.2 
AIC (smaller is better)        2483.2 
AICC (smaller is better)       2483.6 
BIC (smaller is better)        2517.3 

                      Solution for Fixed Effects 
                                Standard 
Effect              Estimate       Error      DF    t Value    Pr > |t| 
Intercept             5.4536      0.3286     192      16.59      <.0001 
time                 -0.8811      0.1039     181      -8.48      <.0001 
agecohort85          -0.2280     0.09910     183      -2.30      0.0225 
deadcohort6          -0.5233     0.09171     175      -5.71      <.0001 
time*agecohort85     0.01805     0.02654    90.3       0.68      0.4980 
time*deadcohort6    -0.06186     0.02663     123      -2.32      0.0218 
 
NOW how do we interpret the fixed intercept and linear slope? 
They are now the expected baseline value and linear rate of change for someone who began the study at 85 and had 
6 years left to live. 
 
 
What do the effects of agecohort85 and deadcohort6 represent? 
The baseline prose recall is .23 lower for every year older than 85 at baseline, and .52 lower for every year 
closer to death at baseline. 
 
 
What do the interactions of agecohort85 and deadcohort6 on the linear time slope represent? 
The linear rate of change over time in prose recall is .02 less negative per year older than 85 at baseline 
(n.s.), and .06 more negative per year closer to death at baseline.  

Given that all models were estimated in ML, one could use AIC and BIC to compare alternative time 
metrics and decide which combination is best empirically. In these data, the models that only 
contain information about age fit less well than those that contain information about years to death. 
 
Sample Description of Results 
 
The extent to which longitudinal change over five occasions in long-term memory as measured by prose recall could 
be described as a function of age (i.e., years since birth), years to death (i.e., negative values up to 0), or time in study 
(i.e., 0 to 8 years, assessments every two years) was evaluated in a sample of 702 older adults. Initial age ranged from 
80 to 98 years; 52% of the variance in age was between-persons, as indicated by an intraclass correlation calculated 
from an empty model for age. Initial years to death ranged from −14 to 0 years; 25% of the variance in years to death 
was between persons. For the outcome of prose recall, 40% of its variance was between persons. 
 
To begin, Figure 1a displays the observed means of prose recall as a function of current age (rounded to the nearest 
two years). As shown, the overall mean age trajectory appears less negative than do the individual age trajectories. A 
random linear model was the best-fitting unconditional model of change as a function of current age (centered such 
that 0 represented 85 years). In that model, as seen in the first columns of Table 1, the expected prose recall at age 85 
was 4.73, with an expected linear rate of decline of 0.51 per additional year of age. However, given that 52% of the 
variance in age was between persons, it is important to examine the extent to which the cross-sectional and 
longitudinal effects of age may differ; only including current age as a predictor assumes they are the same (i.e., 
assumes convergence). Accordingly, age at baseline (also centered at 85 years) was then included as a predictor of 
the intercept and linear age slope to examine the potential for age cohort effects in the level of memory at age 85 and 
the predicted linear rate of change. These model parameters, as seen in the second columns of Table 1, can be 
interpreted as follows. A person who entered the study at age 85 is expected to have an initial prose recall of 5.51 with 
a significant expected linear rate of change per year of −0.68. For every year older than 85 at baseline, prose recall at 
the beginning of the study is expected to be significantly higher by 0.42, and its subsequent rate of linear change per 
year of age is expected to be nonsignificantly more negative by 0.02. Thus, as shown in Figure 1b, although age 
convergence was not obtained for the intercept, indicating positive age cohort differences in expected long-term 
memory at age 85, the expected linear rate of change per year of age was the same across age cohorts. 
 
Next, Figure 2a displays the observed means of prose recall as a function of current years to death (rounded to the 
nearest two years). As shown, the overall mean trajectory still appears less negative than do the individual trajectories, 
although this discrepancy does not appear as substantial as when age was the metric of time. A fixed quadratic, 
random linear model was the best-fitting unconditional model of change as a function of current years to death 
(centered such that 0 represented 6 years prior to death). In that model, as seen in the first columns of Table 2, the 
expected prose recall at 6 years prior to death was 6.02, with a significant expected instantaneous linear rate of 
decline of 0.67 per year closer to death, as evaluated 6 years prior to death. That linear rate of decline is expected to 
become significantly more negative by 0.08 (twice the quadratic coefficient) per year closer to death. However, given 
that 25% of the variance in years to death was between persons, it is important to examine the extent to which the 
cross-sectional and longitudinal effects of years to death may differ; only including current years to death as a predictor 
assumes they are the same (i.e., assumes convergence). Accordingly, years to death at baseline (also centered at 6 
years prior to death) was then included as a predictor of the intercept, linear slope, and quadratic slope to examine the 
potential for death cohort effects in the level of memory at 6 years prior to death and the predicted linear and quadratic 
rates of change as a function of impending death. The parameters from the new model, as seen in the second 
columns of Table 2, can be interpreted as follows. A person who entered the study at 6 years to death is expected to 
have an initial prose recall of 6.37 with a significant expected instantaneous linear rate of change per year closer to 
death of −0.86, which becomes nonsignificantly more negative by 0.03 per year closer to death. For every year closer 
to death at baseline, prose recall at 6 years prior to death is expected to be nonsignificantly higher by 0.17, its rate of 
linear change per year at 6 years prior to death is expected to be nonsignificantly more negative by 0.05, and its rate of 
quadratic change per year is expected to be nonsignificantly less negative by 0.00. Thus, as shown in Figure 2b, 
convergence was obtained for the entire change trajectory; this lack of cohort effects indicates that years to death may 
be an adequate descriptor of change in long-term memory over time.  
 
Finally, Figure 3a displays the observed means of prose recall as a function of years in study (rounded to the nearest 
two years). As before, the overall mean trajectory also appears less negative than do the individual trajectories. A 
random linear model was the best-fitting unconditional model of change as a function of time in study (centered such 
that 0 represented baseline). In that model, as seen in the first columns of Table 3, the expected prose recall at 
baseline was 5.69, with an expected rate of decline of 0.66 per additional year in study. Unlike age and years to death, 
however, time in study has no between-person variation, and thus these are within-person effects. Accordingly, we 

then examined the extent to which individual differences in level of memory at baseline and change in memory during 
the study could be explained by age at baseline (centered at 85 years) or years to death at baseline (centered at 6 
years prior to death). These model parameters, as seen in the second columns of Table 3, can be interpreted as 
follows. A person who entered the study at age 85 and with 6 years until death is expected to have an initial prose 
recall of 5.45 with a significant expected linear rate of change per year of −0.88. For every year older than 85 at 
baseline, prose recall at the beginning of the study is expected to be significantly lower by 0.23, and its subsequent 
rate of linear change per year is expected to be nonsignificantly less negative by 0.02. For every year closer to death 
than 6 years at baseline, prose recall at the beginning of the study is expected to be significantly lower by 0.52, and its 
subsequent rate of linear change per year in study is expected to be significantly more negative by 0.06. Thus, as 
shown in Figure 3b, although older persons have lower prose recall initially, they do not decline at a faster rate. 
Conversely, persons who begin the study closer to death have lower initial prose recall and also decline at a higher 
rate. In terms of effect size, relative to the effect of being an additional year older at baseline, it appears that the effect 
of being an additional year closer to death at baseline is almost twice as large for baseline prose recall and three times 
as large for rate of decline in prose recall. Finally, in comparing all estimated models, the AIC and BIC were smallest 
for the unconditional years to death model (i.e., without the nonsignificant effects of baseline years to death), further 
supporting its utility as a descriptor and potential mechanism of cognitive decline.  
 
(Table 1 would have the two age models, Table 2 would have the two years to death models, and Table 3 would have 
the two time in study models. Figures 1a, 2a, and 3a would show the spaghetti plots for age, years to death, and time 
in study, respectively, whereas Figures 1b, 2b, and 3b, would show the conditional model predictions for age, years to 
death, and time in study, respectively.) 
 
 


